*------------------------------------------------------------------------------*
* Family history - Dinas, Fouka, Schläpfer - JoP
*------------------------------------------------------------------------------*

*------------------------------------------------------------------------------*
* Figure 3
*------------------------------------------------------------------------------*



*------------------------------------------------------------------------------*
* Greece
*------------------------------------------------------------------------------*

use greece, clear

gen inter=asiaminor*T
local controls female i.age i.edu i.income i.occupation i.nomos 
local sample if T==0

estimates clear
eststo att: reg pc_all_bst asiaminor `controls' `sample', cl(municipality)
eststo beh: reg pc_behst asiaminor `controls' `sample', cl(municipality)


coefplot (beh, offset(-0.01) aseq("Quasi-behavioral") ciopts(lcolor(blue blue)) mcolor(blue)) (att, aseq("Attitudes") ciopts(lcolor(red red)) mcolor(red)), ///
levels(95 90) xline(0, lcolor(gs10)) keep(asiaminor) ///
ysca(titlegap(2)) msize(small) title("Greece", size(small)) ///
xtitle("", size(small)) xtitle("Baseline difference", size(small)) ///
xlabel(, nogrid labsize(small)) ylabel(, nogrid labsize(small)) xsca(titlegap(2)) ///
legend(region(lcolor(white))) swapnames legend(off) grid(none) 

graph save Greece, replace


*------------------------------------------------------------------------------*
* Germany
*------------------------------------------------------------------------------*

use germany, clear

local controls i.sex i.state i.age i.F4 i.F13
estimates clear
eststo att: reg pc_att expellee `controls' if control==1, cl(GKZ)
eststo beh: reg donate_st expellee `controls' if control==1, cl(GKZ)


coefplot (beh, offset(-0.01) aseq("Donation UNHCR") ciopts(lcolor(blue blue)) mcolor(blue)) (att, aseq("Attitudes") ciopts(lcolor(red red)) mcolor(red)), ///
levels(95 90) xline(0, lcolor(gs10)) keep(expellee) ///
ysca(titlegap(2)) msize(small) title("Germany", size(small)) ///
ytitle("", size(small)) xtitle("Baseline difference", size(small)) ///
xlabel(, nogrid labsize(small)) ylabel(, nogrid labsize(small)) xsca(titlegap(2)) ///
legend(region(lcolor(white))) swapnames legend(off) grid(none) 

graph save Germany, replace

*------------------------------------------------------------------------------*
* Combine
*------------------------------------------------------------------------------*

graph combine Greece.gph Germany.gph, xcommon iscale(2) ysize(2)
